

## Table des matières

<b>1</b>	<b>Préambule</b>	<b>2</b>
<b>2</b>	<b>Faire un plot 2D</b>	<b>2</b>
2.1	À partir d'un fichier . . . . .	2
2.2	Tracer une fonction à partir de la formule . . . . .	2
<b>3</b>	<b>Faire un plot 3D</b>	<b>3</b>
3.1	Surface plot . . . . .	3
3.1.1	Modifier la palette de couleurs . . . . .	3
3.1.2	Ajouter des contours à une surface . . . . .	3
<b>4</b>	<b>Exporter la courbe</b>	<b>5</b>
4.1	Fichier pdf . . . . .	5
4.2	fichier svg . . . . .	5
4.3	fichier png . . . . .	6
<b>5</b>	<b>Les options</b>	<b>6</b>
5.1	label . . . . .	6
5.2	échelle logarithmique . . . . .	6
5.3	grille . . . . .	6
5.4	Forcer un zoom . . . . .	6
5.5	Style de courbe . . . . .	6
5.6	Texte avancé . . . . .	6

# 1 Préambule

Gnuplot permet de tracer des courbes de manière plus ou moins complexes, en 2D ou en 3D. Ce formulaire est là pour donner des pistes et pouvoir faire rapidement des choses basiques. Il n'a pas la prétention d'être exhaustif, ce qui rendrait l'information plus difficile à retrouver et à séparer de l'essentiel que je m'efforce de mettre en valeur.

Gnuplot se présente sous la forme d'un logiciel en ligne de commande que l'on utilise en faisant :

```
gnuplot
```

Toutes les commandes que je présente peuvent être entrées de manière interactive dans **gnuplot**. Je ne présenterais pas cette manière de faire, dans la suite je présenterais une suite d'instruction que j'enregistre dans un fichier (au nom quelconque, mais à qui je donne l'extension **.gnuplot** pour plus de lisibilité. Afin de s'en servir on fait alors :

```
gnuplot fichier.gnuplot
```

**Remarque** : La différence principale entre le script **\*.gnuplot** et l'affichage interactif, c'est que le plot affiché à l'écran (terminal **x11**) ne reste pas par défaut, il faut ainsi mettre un **pause(-1)** en fin de script pour que le graphique reste, et un appui sur Entrée continue l'exécution du script (et le termine s'il n'y a rien ensuite).

## 2 Faire un plot 2D

### 2.1 À partir d'un fichier

Admettons que le fichier **plot.dat** contienne deux colonnes dans lesquelles il y a respectivement les valeurs de  $x$  et  $y$  pour notre graphique. Afin d'afficher le graphique il suffit de faire :

```
plot 'plot.dat' using 1:2
```

On va alors tracer la deuxième colonne en fonction de la première.

On peut aussi tracer plusieurs courbes :

```
plot 'plot.dat' using 1:2 title 'courbe1', '' using 1:3 title 'courbe2', \
'' using 1:4 title 'courbe3', 'plot2.dat' using 1:2 title 'courbe4'
```

Pour tracer 4 courbes, 3 venant du fichier **plot.dat**, la 2<sup>e</sup>, 3<sup>e</sup> et 4<sup>e</sup> colonne en fonction de la première et une courbe venant de **plot2.dat**.

### 2.2 Tracer une fonction à partir de la formule

On peut, au lieu de rentrer un tableau de points, tracer directement une fonction connue :

```
plot sin(x)
```

Les fonctions dans gnuplot sont les mêmes que les fonctions dans la librairie mathématique Unix à part que les fonctions acceptent des entiers des réels ou des complexes comme argument.

**Remarque** : Pour les fonctions qui acceptent ou retournent des angles qui pourraient être donnés en degrés ou en radian ( $\sin(x)$ ,  $\cos(x)$ ,  $\tan(x)$ ,  $\arcsin(x)$ ,  $\arccos(x)$ ,  $\arctan(x)$ , ...) l'unité peut être sélectionnée par **set angles** (par défaut, ce sont des radians).

On peut alors faire par exemple :

```
plot (sin(x)/x)**2
```

un sinus cardinal au carré.

Si le nombre de points n'est pas assez élevé, on peut alors rajouter

```
set samples 200
```

où 200 est le nombre de points servant à tracer la fonction. Si vous êtes en interactif ou que l'option **samples** est utilisée après le **plot**, il faut aussi utiliser

```
replot
```

## 3 Faire un plot 3D

### 3.1 Surface plot

Pour afficher un graphique sous la forme d'une surface (x,y) dont la couleur dépend de z, il faut d'une part préparer le fichier de données de manière spéciale. Les données doivent être sous la forme :

```
# X      Y      Z
0.0    0.0    2.0
0.0    1.0    2.0
0.0    1.0    3.5
0.0    2.0    3.5

1.0    0.0    2.0
1.0    1.0    2.0
1.0    1.0    3.5
1.0    2.0    3.5

1.0    0.0    1.0
1.0    1.0    1.0
1.0    1.0    3.0
1.0    2.0    3.0

2.0    0.0    1.0
2.0    1.0    1.0
2.0    1.0    3.0
2.0    2.0    3.0
```

 Il faut séparer les blocs de  $X = c^{te}$  par une ligne blanche sans quoi **gnuplot** ne fonctionnera pas.

Ensuite, il faut utiliser les commandes suivantes :

```
set terminal x11 enhanced
set xlabel "semi major axis (AU)"
set ylabel "Planet mass (m_{earth})"
set title "Evolution of the total torque {/Symbol G}"
set pm3d map
splot 'test_total_torque.dat' title ''
```

dans **splot**, le titre vide est là pour ne pas avoir le nom du fichier imprimé au milieu de la courbe. **map** est l'option qui permet d'avoir une carte au lieu d'avoir le graphique 3D. Il est aussi possible d'obtenir la même chose en utilisant l'option qui permet de changer l'angle de visée sous lequel regarder le graphe 3D **set view map; unset surface**

**Remarque :** Je n'ai pas réussi à imposer, via **set {x|y}range** le zoom du graphique. Dans les bords, en particulier quand l'échantillonnage est faible, ça peut ne pas être très joli.

#### 3.1.1 Modifier la palette de couleurs

Afin de changer la palette de couleurs, et ainsi avoir de meilleurs contrastes avec les couleurs proches, on peut faire

```
set palette rgbformulae 22,13,-31
```

et avoir une sorte d'arc en ciel.

#### 3.1.2 Ajouter des contours à une surface

Ci-dessous voici le contenu d'un script que j'utilise pour afficher un contour sur l'iso-ligne  $z = 0$  :

```

set contour base; set cntrparam levels discret 0.
set isosamples 100, 100
unset surface
set table "contour.dat"
set dgrid3d
splot 'test_total_torque.dat'
unset table
# Change single blank lines to double blank lines
!awk "NF<2{printf\"\n\"}{print}" <contour.dat >contour1.dat
# Draw the plot
reset
set terminal x11 enhanced
set xlabel "semi major axis (AU)"
set ylabel "Planet mass (m_{earth})" center
set title "Evolution of the total torque {/Symbol G}_{tot} "
set pm3d map
set pm3d explicit
set palette rgbformulae 22,13,-31
splot 'test_total_torque.dat' with pm3d title '',\
      'contour1.dat' with line lt -1 title ''
pause -1 # wait until a carriage return is hit
set terminal pngcairo enhanced
set output 'total_torque.png'
replot # pour générer le fichier d'output
!rm contour.dat contour1.dat

```

```

set contour base; set cntrparam levels discret 0.

```

On définit les niveaux des contours (ici, uniquement une ligne pour  $z = 0$ ).

```

set isosamples 100, 100

```

On définit l'échantillonnage, 100 points pour  $x$  et  $y$  (sinon les lignes ne sont pas suffisamment bien résolues. On peut bien entendu encore augmenter ce nombre.

```

unset surface
set table "contour.dat"
set dgrid3d
splot 'test_total_torque.dat'
unset table
# Change single blank lines to double blank lines
!awk "NF<2{printf\"\n\"}{print}" <contour.dat >contour1.dat
# Draw the plot
reset

```

Cette partie là est plus ou moins une boîte noire. La ligne `set dgrid3d` dans le cas d'un fichier de données (contrairement à une fonction que l'on définirait directement dans gnuplot ( $f(x, y) = x * x - y * y$  par exemple). On stocke alors les valeurs des contours dans un fichier **contour.dat**, puis **contour1.dat** (pour un fichier qui présente des doubles lignes, apparemment nécessaires à l'affichage correct des contours pour la suite).



Les contours seront ceux du fichier **test\_total\_torque.dat**, celui-ci doit bien évidemment être le même que celui qui sera affiché en même temps que le contour dans la 2<sup>e</sup> partie du script, sinon ça n'a pas de sens.

```

set terminal x11 enhanced
set xlabel "semi major axis (AU)"
set ylabel "Planet mass (m_{earth})" center
set title "Evolution of the total torque {/Symbol G}_{tot} "

```

On définit le terminal et les titres. Les labels ne sont bien entendu pas obligatoire.

```
set pm3d map
set pm3d explicit
```

On définit le fait qu'on peut un plot 3D, affiché uniquement en 2D (avec le  $z$  variant selon une palette de couleur. On veut que les plots soient explicitement spécifiés comme 3D, ceci nous permet d'afficher les contours simplement en 3D (des lignes noires quoi).

```
set palette rgbformulae 22,13,-31
```

On change la palette de couleur (c'est un arc en ciel, plus pratique pour discerner les couleurs, et donc les valeurs associées.

```
splot 'test_total_torque.dat' with pm3d title '', \
      'contour1.dat' with line lt -1 title ''
```

On affiche les données de **test\_total\_torque.dat** et le contour **contour1.dat** (qui est censé correspondre à **test\_total\_torque.dat** sinon ça n'a pas de sens). La partie **title ''** permet d'éviter l'affichage du nom du fichier sur la courbe.

```
pause -1 # wait until a carriage return is hit
```

Le graphique restera affiché à l'écran tant qu'on n'a pas appuyé sur la touche Entrée (et ainsi continuer le script)

```
set terminal pngcairo enhanced
set output 'total_torque.png'
replot # pour générer le fichier d'output
```

On stocke le graphique dans un fichier **.png**

```
!rm contour.dat contour1.dat
```

On efface les fichiers temporaires qui ont été créés pour les contours.

## 4 Exporter la courbe

Le principe de gnuplot est de sélectionner un terminal où envoyer les données du graphique. par défaut, ce terminal est **x11**.

Pour définir un terminal il suffit de rentrer

```
set terminal x11
```

Cette dernière commande est la commande par défaut, celle qui permet d'afficher les graphiques à l'écran.

Dans le mode interactif de gnuplot, on accède à la liste des terminaux via

```
help terminal
```

### 4.1 Fichier pdf

```
set terminal pdfcairo
set output 'graphique.pdf'
replot
```

**Remarque :** Le **replot** n'est utile que si la courbe a déjà été affichée (à l'écran par exemple).

### 4.2 fichier svg

```
set terminal svg rounded size 450,360
set output 'graphique.svg'
replot
```

## 4.3 fichier png

```
set terminal pngcairo
set output 'graphique.png'
replot
```

## 5 Les options

### 5.1 label

```
set xlabel "axe des abscisses"
set ylabel "axe des ordonnées"
```

### 5.2 échelle logarithmique

```
set logscale x
set logscale y
```

### 5.3 grille

```
set grid
```

affiche une grille sur les traits majeurs.

Pour afficher aussi la grille mineure, il faut utiliser à la place

```
set mxtics 5
set mytics 5
set grid xtics ytics mxtics mytics
```

**Remarque :** `set mxtics 5` signifie que la grille mineure horizontale va subdiviser un grand intervalle en 5 sous intervalles. Si l'intervalle est de 0 à 10, il y aura donc des traits mineurs pour 2, 4, 6 et 8.

### 5.4 Forcer un zoom

```
set xrange [0:5]
```

### 5.5 Style de courbe

On peut forcer l'utilisation de lignes (par exemple) pour afficher les courbes :

```
plot 'test_opacity.dat' using 1:2 with lines
```

### 5.6 Texte avancé

Dans le terminal<sup>1</sup> courant, lors de sa définition, il faut rajouter l'option **enhanced**

```
set terminal x11 enhanced
```



Tous les terminaux ne supportent l'affichage de texte avancé (et donc l'option **enhanced**)

Ceci permet de faire du texte en indice, en exposant et d'accéder aux lettres grecques par exemple. On peut alors écrire dans le graphique  $x_1 + y^2$  qui apparaîtra comme  $x_1 + y^2$ .

**Remarque :** Pas besoin de mettre en mode mathématique, il suffit de l'écrire comme ça.

Une différence importante a lieu pour l'affichage des lettres grecques :


---

1. Pour plus de détails sur la définition du terminal, voir [§ 4]

ALPHABET	SYMBOL	ALPHABET	SYMBOL	alphabet	symbol	alphabet	symbol
A	A	N	N	a	$\alpha$	n	$\nu$
B	B	O	O	b	$\beta$	o	o
C	X	P	$\Pi$	c	$\chi$	p	$\pi$
D	$\Delta$	Q	$\Theta$	d	$\delta$	q	$\theta$
E	E	R	P	e	$\epsilon$	r	$\rho$
F	$\Phi$	S	$\Sigma$	f	$\phi$	s	$\sigma$
G	$\Gamma$	T	T	g	$\gamma$	t	$\tau$
H	H	U	Y	h	$\eta$	u	$v$
I	I	V	$\varsigma$	i	$\iota$	v	$\varpi$
J	$\vartheta$	W	$\Omega$	j	$\varphi$	w	$\omega$
K	K	X	$\Xi$	k	$\kappa$	x	$\xi$
L	$\Lambda$	Y	$\Psi$	l	$\lambda$	y	$\psi$
M	M	Z	Z	m	$\mu$	z	$\zeta$

Ainsi, pour faire la lettre minuscule  $\alpha$  il faut faire :

`{/Symbol a}`

 Ce n'est pas un anti-slash comme les habitudes L<sup>A</sup>T<sub>E</sub>X pourraient nous le faire croire.